
APPENDIX 3:_MPLUS SCRIPT FOR ACE MODEL WITH GXE INVOLVING AN INDIVIDUAL-LEVEL ENVIRONMENTAL VARIABLE (FIGURE 3)
title: 

    ACE model with GxE involving an individual-level environmental variable
    see figure 3)

data:

    file is RD1.dat;

variable:

    names are famnum zyg

              reject1 reject2 

              dep1 dep2;

    missing are all (99);                  !   Specify missing data symbol.

    usevariables are y1_1 y2_1 y1_2 y2_2;

    constraint = m1 m2;

    grouping is zyg (1=mz 2=dz);           !   Identify the grouping variable.

define:

    !

    !   Define the variables to be analyzed.

    !   where

    !   yi_j : phenotype i twin j

    !

    !   mj : Moderator variable twin j

    !

    y1_1=reject1;

    y2_1=dep1;

    !

    y1_2=reject2;

    y2_2=dep2;

    !

    m1=reject1;

    m2=reject2;

analysis:

    type = general;

model:

    !

    !   The overall model.

    !   We estimate:

    !

    [y1_1]; [y1_2];     !   Means

    [y2_1]; [y2_2];

    !

    y1_1; y1_2;         !   Variances

    y2_1; y2_2;

    !

    y2_1 with y1_1;     !   Covariances

    !

    y2_2 with y1_2;

    !

    y1_2 with y1_1;

    y2_2 with y2_1;

    !

    y2_2 with y1_1;

    y2_1 with y1_2;

model mz:

    !

    !   The specific model for each group.

    !   Assign names to the parameters.

    !   The parameter names are provided in parentheses.

    !   Same name = equal parameters.

    !

    [y1_1](m1mz); [y1_2](m1mz);     !   Means of phenotype 1 are equal across twins. 

    [y2_1](m2); [y2_2](m2);         !   Means of phenotype 2 are equal across twins 

                                    !   and across MZ and DZ.

    !

    !   The variance-covariance structure depends on the moderator variable 

    !   (see model constraint section).

    !

    y1_1(t11); y1_2(t11);           

    y2_1(t22_1); y2_2(t22_2);       

    !

    y2_1 with y1_1(t21_1);

    !

    y2_2 with y1_2(t21_2);

    !

    y1_2 with y1_1(b11mz);

    y2_2 with y2_1(b22mz);

    !

    y2_2 with y1_1(b21mz);

    y2_1 with y1_2(b12mz);

model dz:

    [y1_1](m1dz); [y1_2](m1dz);

    [y2_1](m2); [y2_2](m2);

    !

    y1_1(t11); y1_2(t11);

    y2_1(t22_1); y2_2(t22_2);

    !

    y2_1 with y1_1(t21_1);

    !

    y2_2 with y1_2(t21_2);

    !

    y1_2 with y1_1(b11dz);

    y2_2 with y2_1(b22dz);

    !    

    y2_2 with y1_1(b21dz);

    y2_1 with y1_2(b12dz);

model constraint:

    !

    !   Constraints applied to the model.

    !

    new(ar ac au cr cc cu er ec eu

        bac bau bcc bcu bec beu);

    !

    !   Within-twin variances and covariances

    !   Except for the variance of phenotype 1(t11), all other variances and 

    !   covariances are a function of the moderator variable(m).

    !

    t11=ar**2 + cr**2 + er**2;

    t21_1=ar*(ac+bac*m1) + cr*(cc+bcc*m1) + er*(ec+bec*m1);

    t21_2=ar*(ac+bac*m2) + cr*(cc+bcc*m2) + er*(ec+bec*m2);

    !

    t22_1=(ac+bac*m1)**2 + (au+bau*m1)**2 + 

          (cc+bcc*m1)**2 + (cu+bcu*m1)**2 + 

          (ec+bec*m1)**2 + (eu+beu*m1)**2;

    t22_2=(ac+bac*m2)**2 + (au+bau*m2)**2 + 

          (cc+bcc*m2)**2 + (cu+bcu*m2)**2 + 

          (ec+bec*m2)**2 + (eu+beu*m2)**2;

    !

    !   Between-twin covariances.

    !   Except for the covariance between twin 1 and twin 2 of phenotype 1 (b11mz 

    !   and b11dz), all other covariances are a function of the moderator variable(m). 

    !

    b11mz=ar**2 + cr**2;

    b21mz=ar*(ac+bac*m2) + cr*(cc+bcc*m2);

    b12mz=ar*(ac+bac*m1) + cr*(cc+bcc*m1);

    !

    b22mz=(ac+bac*m1)*(ac+bac*m2) + 

          (au+bau*m1)*(au+bau*m2) + (cc+bcc*m1)*(cc+bcc*m2) + 

          (cu+bcu*m1)*(cu+bcu*m2);

    b11dz=0.5*ar**2 + cr**2;

    b21dz=0.5*ar*(ac+bac*m2) + cr*(cc+bcc*m2);

    b12dz=0.5*ar*(ac+bac*m1) + cr*(cc+bcc*m1);

    b22dz=0.5*((ac+bac*m1)*(ac+bac*m2) + (au+bau*m1)*(au+bau*m2)) + 

          (cc+bcc*m1)*(cc+bcc*m2) + (cu+bcu*m1)*(cu+bcu*m2);

output:

    cinterval;      !   Request 95% and 99% confidence intervals.
